

# Séries de Fourier

## Fonctions définies par morceaux

### Définition

Il arrive fréquemment qu'en théorie des séries de Fourier, on soit amené à manipuler des fonctions définies par morceaux. Maple permet de définir de telles fonctions grâce à la procédure **piecewise** dont la syntaxe est

`piecewise(condition1,valeur1,condition2,valeur2,...,conditionN,valeurN,valeur_sinon)`  
où conditionI est une condition et valeurI une valeur. Dès qu'une condition est remplie (de gauche à droite), Maple renvoie la valeur correspondante. Si aucune de ces conditions n'est vérifiée, Maple renvoie la dernière valeur, que nous avons dénotée par valeur\_sinon.

> **f:=x->piecewise(x<-2,x+2,x<1,1/2,x<2,x^2/2,x-1);**

$$f := x \quad \text{piecewise} \quad x < -2, x + 2, x < 1, \frac{1}{2}, x < 2, \frac{1}{2}x^2, x - 1$$


avec une représentation à l'écran assez explicite

> **f(x);**

$$\begin{array}{ll} x + 2 & x < -2 \\ \frac{1}{2} & x < 1 \\ \frac{1}{2}x^2 & x < 2 \\ x - 1 & \text{otherwise} \end{array}$$


et une représentation graphique classique

> **plot(f,-3..3);**




Une autre manière de définir des fonctions par morceaux est d'utiliser la fonction de Heaviside, qui vaut 0 pour  $x < 0$  et 1 pour  $x > 0$  (sa valeur au point 0 est indéterminée).

> **plot(Heaviside,-2..2);**



```
> plot(Heaviside(-x)*(x+1)+Heaviside(x-2)*(x-3)+1,x=-3..3);
```



On peut librement convertir entre les fonctions piecewise et les fonctions définies avec la fonction de Heaviside à l'aide de la fonction **convert**.

```
> convert(piecewise(x<-2,x+2,x<1,1/2,x<2,x^2/2,x-1),Heaviside);
```

$$x - \frac{1}{2} \text{Heaviside}(x - 1) + 2 - \frac{3}{2} \text{Heaviside}(x + 2) - \text{Heaviside}(x - 2) \\ - x \text{Heaviside}(x + 2) + \frac{1}{2} x^2 \text{Heaviside}(x - 1) - \frac{1}{2} x^2 \text{Heaviside}(x - 2) \\ + \text{Heaviside}(x - 2) x$$

```
> convert(Heaviside(-x)*(x+1)+Heaviside(x-2)*(x-3)+1,piecewise);
```

$$\begin{array}{ll} x + 2 & x < 0 \\ \text{undefined} & x = 0 \\ 1 & 0 < x < 2 \\ \text{undefined} & x = 2 \\ x - 2 & 2 < x \end{array}$$

## Manipulation

Les fonctions définies par morceaux peuvent être dérivées, intégrées et même utilisées dans des équations différentielles. Pour dériver, nous conseillons d'introduire la fonction sous forme **piecewise**, car la fonction de Heaviside n'étant pas dérivable au sens classique (sa *dérivée* est une distribution et pas une fonction), les résultats peuvent être réellement difficiles à interpréter.

Exemple de dérivation:

```
> f:=x->piecewise(x<-2,x+2,x<1,1/2,x<2,x^2/2,x-1):
diff(f(x),x);
```

```

1      x < -2
undefined x = -2
0      x < 1
undefined x = 1
x      x < 2
undefined x = 2
1      2 < x

```

Exemple de primitivation:

```
> int(f(x), x);
```

```

1/2 x^2 + 2 x      x -2
-1 + 1/2 x      x 1
-2/3 + 1/6 x^3      x 2
-x + 2/3 + 1/2 x^2      2 < x

```

Exemple d'intégration:

```
> int(f(x), x=0..2*Pi);
```

```

5/3 -2 + 2^2

```

Exemple de résolution d'équation différentielle:

```
> dsolve(diff(y(x), x)+y(x)=f(x), y(x));
```

```

x + 1 + e^(-x) _CI      x -2
1/2 - 3/2 e^(-2 -x) + e^(-x) _CI      x 1
y(x) = 1 - x - 3/2 e^(-2 -x) + 1/2 x^2 + e^(-x) _CI      x 2
e^(2 -x) - 2 + x - 3/2 e^(-2 -x) + e^(-x) _CI      2 < x

```

```
>
```

## ☐ Coefficients et séries de Fourier

### ☐ Coefficients de Fourier

Maple ne possède pas de fonction spécialisée dans le calcul des coefficients de Fourier, mais il n'est pas bien difficile d'y remédier en calculant les intégrales

$$c_n(f) = \frac{1}{T} \int_a^{a+T} f(x) e^{-\frac{2\pi i n x}{T}} dx, \quad a_n(f) = \frac{1}{T} \int_a^{a+T} f(x) \cos \frac{2\pi n x}{T} dx$$

$$b_n(f) = \frac{1}{T} \int_a^{a+T} f(x) \sin \frac{2\pi n x}{T} dx$$

et

$$b_n(f) = \frac{1}{T} \int_a^{a+T} f(x) \sin \frac{2\pi n x}{T} dx$$

de la manière suivante

```

> Fourier_c:=(f,x,n,a,T)->int(f*exp(-2*Pi*I*n*x/T),x=a..a+T)/T;
Fourier_a:=(f,x,n,a,T)->2*int(f*cos(2*Pi*n*x/T),x=a..a+T)/T;
Fourier_b:=(f,x,n,a,T)->2*int(f*sin(2*Pi*n*x/T),x=a..a+T)/T;

```

$$f e^{-2 \frac{I}{T} n x} dx = a .. a + T$$

$$Fourier\_c := (f, x, n, a, T) \frac{\int_a^{a+T} f e^{-2 \frac{I}{T} n x} dx}{T}$$

$$f \cos 2 \frac{n x}{T} dx = a .. a + T$$

$$Fourier\_a := (f, x, n, a, T) \frac{2 \int_a^{a+T} f \cos 2 \frac{n x}{T} dx}{T}$$

$$f \sin 2 \frac{n x}{T} dx = a .. a + T$$

$$Fourier\_b := (f, x, n, a, T) \frac{2 \int_a^{a+T} f \sin 2 \frac{n x}{T} dx}{T}$$

On peut même faire mieux en terme de procédures en donnant des valeurs par défaut à la période (2 par exemple) et au paramètre de début d'intégration, en utilisant les fonctionnalités de Maple concernant le passage des paramètres (voir dans l'aide **nargs** et **args**):

```
> Fourier_c:=proc(f,x,n) local a,T;
    if nargs<5 then T:=2*Pi else T:=args[5] fi;
    if nargs<4 then a:=0 else a:=args[4] fi;
    int(f*exp(-2*Pi*I*n*x/T),x=a..a+T)/T
end:
> Fourier_a:=proc(f,x,n) local a,T;
    if nargs<5 then T:=2*Pi else T:=args[5] fi;
    if nargs<4 then a:=0 else a:=args[4] fi;
    2*int(f*cos(2*Pi*n*x/T),x=a..a+T)/T
end:
> Fourier_b:=proc(f,x,n) local a,T;
    if nargs<5 then T:=2*Pi else T:=args[5] fi;
    if nargs<4 then a:=0 else a:=args[4] fi;
    2*int(f*sin(2*Pi*n*x/T),x=a..a+T)/T
end:
```

Période 2 avec début en 0:

```
> Fourier_c(g(x),x,n);
```

$$\frac{1}{2} \int_0^2 g(x) e^{(-I n x)} dx$$

Période 2 avec début en -

```
> Fourier_c(g(x),x,n,-Pi);
```

$$\frac{1}{2} \int_{-\pi}^{\pi} g(x) e^{(-I n x)} dx$$

Période 1 avec début en 0

```
> Fourier_c(g(x),x,n,0,1);
```

$$\int_0^1 g(x) e^{(-2 I n x)} dx$$

Avec quelques exemples de calculs:

```
> Fourier_c(x,x,n,-Pi);
```

$$\frac{1}{2} \frac{(I n + 1) e^{(-I n)} + e^{(I n)} (I n - 1)}{n^2}$$

On voit que l'on doit simplifier car on n'a pas dit à Maple que n est un entier:

```
> assume(n, integer): simplify(");
```

$$\frac{1}{2} \frac{I(n e^{(-I n)} - I e^{(-I n)} + n e^{(I n)} + I e^{(I n)})}{n^2}$$

ce qui n'arrange guère les choses: Maple dans cette version ne sait pas que  $e^{(In)} = (-1)^n$ . Essayons avec les fonctions réelles:

```
> Fourier_a(x,x,n,-Pi);
```

$$0$$

```
> Fourier_b(x,x,n,-Pi);
```

$$-2 \frac{(-1)^n}{n}$$

cela marche nettement mieux.

Prenons maintenant une fonction définie par morceaux: la fonction définie par  $f(x)=0$  sur  $[0, \pi]$  et  $x^2$  sur  $[\pi, 2\pi]$  et périodique de période  $2\pi$ .

```
> f:=x->piecewise(x<0,0,x^2);
```

$$f:=x \quad \text{piecewise}(x < 0, 0, x^2)$$

```
> Fourier_c(f(x),x,n,-Pi);
```

$$\frac{1}{2} \frac{I^2 n^2 + 2 I (-1)^n - 2 I + 2 n}{(-1)^n n^3}$$

```
> Fourier_a(f(x),x,n,-Pi);
```

$$2 \frac{(-1)^n}{n^2}$$

```
> Fourier_b(f(x),x,n,-Pi);
```

$$-\frac{(-1)^n n^2 - 2 (-1)^n + 2}{n^3}$$

## ■ Séries de Fourier

En utilisant les fonctions précédentes, il est facile de construire une fonction qui calcule soit les sommes partielles de la série de Fourier (sous forme réelle ou complexe), soit la série de Fourier (également sous forme réelle ou complexe); l'usage de la fonction **Sum** à la place de **sum** enlève à Maple toute velléité de calculer de manière explicite les sommes correspondantes, ce qui n'est pas le but recherché

```
> SP_Fourier_comp:=proc(f,x,n) local a,T,k;
    assume(k, integer);
    if nargs<5 then T:=2*Pi else T:=args[5] fi;
    if nargs<4 then a:=0 else a:=args[4] fi;
    Sum(Fourier_c(f,x,k,a,T)*exp(2*Pi*I*k*x/T), k=-n..n)
end:
SP_Fourier_reel:=proc(f,x,n) local a,T,k;
    assume(k, integer);
    if nargs<5 then T:=2*Pi else T:=args[5] fi;
    if nargs<4 then a:=0 else a:=args[4] fi;
    Fourier_a(f,x,0,a,T)/2
    +Sum( Fourier_a(f,x,k,a,T)*cos(2*Pi*k*x/T)
        +Fourier_b(f,x,k,a,T)*sin(2*Pi*k*x/T)
        ,k=1..n)
    end:
> SP_Fourier_comp((Pi-x)/2,x,n);
```

$$\sum_{k=-n}^n \frac{1}{2} \frac{I k + 1}{((-1)^k)^2 k^2} - \frac{1}{2} \frac{I k - 1}{k^2} e^{(I k x)}$$

```
> SP_Fourier_reel((Pi-x)/2,x,n);
```

$$\sum_{k=1}^n \frac{\sin(kx)}{k}$$

Quant à la série de Fourier proprement dite, il suffit de remplacer n par l'infini

```
> S_Fourier_comp:=proc(f,x) local a,T;
  if nargs<4 then T:=2*Pi else T:=args[4] fi;
  if nargs<3 then a:=0 else a:=args[3] fi;
  SP_Fourier_comp(f,x,infinity,a,T)
end:
S_Fourier_reel:=proc(f,x) local a,T;
  if nargs<4 then T:=2*Pi else T:=args[4] fi;
  if nargs<3 then a:=0 else a:=args[3] fi;
  SP_Fourier_reel(f,x,infinity,a,T)
end:
```

```
> S_Fourier_reel((Pi-x)/2,x);
```

$$\sum_{k=1}^{\infty} \frac{\sin(kx)}{k}$$

```
> eval("");
```

$$\sum_{k=1}^{\infty} \frac{\sin(kx)}{k}$$

```
> lprint("");
```

```
Sum(1/k*sin(k*x),k = 1 .. infinity)
```

## ☐ Sommation de séries

### ☐ Théorème de Dirichlet

Le théorème de Dirichlet s'exprime simplement à l'aide des fonctions précédentes et conduit à la sommation de certaines séries (à supposer que les conditions d'application soient vérifiées); il faut veiller à ramener le point dans l'intervalle  $[a, a + T]$  et à distinguer suivant que le point est une extrémité ou non de l'intervalle

```
> Dirichlet:=proc(f,x,x0) local valeur,a,T,x1;
  if nargs<5 then T:=2*Pi else T:=args[5] fi;
  if nargs<4 then a:=0 else a:=args[4] fi;
  x1:=x0-floor((x0-a)/T)*T;
  if x1=a
  then valeur:=(limit(f,x=a+T,left)+
    limit(f,x=a,right))/2
  else valeur:=(limit(f,x=x1,left)+
    limit(f,x=x1,right))/2
  fi;
  valeur= simplify(subs(x=x1,S_Fourier_reel(f,x,a,T)))
end:
```

```
> Dirichlet(x^2,x,Pi);
```

$$\frac{4}{3} + \sum_{k=1}^{\infty} \frac{4(-1)^k}{k^2}$$

```
> Dirichlet(x^2,x,0);
```

$$\frac{4}{3} + \sum_{k=1}^{\infty} \frac{4}{k^2}$$

### ☐ Théorème de Parseval

Le théorème de Parseval s'exprime simplement à l'aide des fonctions précédentes et conduit à la sommation de certaines séries. Nous nous limiterons à des fonctions à valeurs

réelles de telle sorte que Maple soit à même de calculer l'intégrale définissant la norme de la fonction.

```
> Parseval:=proc(f,x) local valeur,a,T,x1,k;
    if nargs<4 then T:=2*Pi else T:=args[4] fi;
    if nargs<3 then a:=0 else a:=args[3] fi;
    assume(k,integer);

    simplify(int(f^2,x=a..a+T)/T=Fourier_a(f,x,0,a,T)^2/4
        +Sum(Fourier_a(f,x,k,a,T)^2+
            Fourier_b(f,x,k,a,T)^2,k=1..infinity)/2)
    end:
>
> Parseval(x^4,x,-Pi);
```

$$\frac{1}{9} \cdot 8 = \frac{1}{25} \cdot 8 + \frac{1}{2} \sum_{k=1}^{\infty} 64 \frac{(k^2 - 6)^2}{k^8}$$

## Convolution des fonctions

La convolution de deux fonctions périodiques de période T est définie par

$$(f@g)(x) = \frac{1}{T} \int_a^{a+T} f(x-t)g(t) dt$$
. Le seul point délicat est de bien ramener les variables à l'intérieur de l'intervalle  $[a, a+T]$  par des translations de kT. On utilise toujours les valeurs par défaut a=0 et T=2

```
> convol:=proc(f,g,t,x) local a,T,fx1,fx2,N;
    if nargs<6 then T:=2*Pi else T:=args[6] fi;
    if nargs<5 then a:=0 else a:=args[5] fi;
    fx1:=subs(t=x-t-N*T,f);
    fx2:=subs(t=x-t-N*T+T,f);
    subs(N=floor((x-2*a)/T),


    (int(fx1*g,t=a..x-a-N*T)+int(fx2*g,t=x-a-N*T..a+T))/T)
    end:
> f:=x->piecewise(x<Pi,0,1); g:=convol(f(t),f(t),t,x);
    f:=x piecewise(x<,0,1)
```

$$g := \frac{1}{2} \int_0^{x-2\%1} \left( \begin{cases} 0 & x-t-2\%1 < \\ 1 & \text{otherwise} \end{cases} \right) \left( \begin{cases} 0 & t < \\ 1 & \text{otherwise} \end{cases} \right) dt$$

$$+ \int_{x-2\%1}^0 \left( \begin{cases} 0 & x-t-2\%1+2 < \\ 1 & \text{otherwise} \end{cases} \right) \left( \begin{cases} 0 & t < \\ 1 & \text{otherwise} \end{cases} \right) dt /$$

$$\%1 := \text{floor} \frac{1}{2} x$$

```
> plot(g,x=0..3*Pi);
```





# Cours de mathématiques

par Denis Monasse

Ed. Vuibert

## Table des matières

- Plan général
- Algèbre générale
- Algèbre linéaire
- Réduction des endomorphismes
- Topologie des espaces métriques
- Espaces vectoriels normés
- Comparaison des fonctions
- Suites et séries numériques
- Fonctions d'une variable réelle
- Intégration
- Suites et séries de fonctions

- Séries entières
- Formes quadratiques
- Formes hermitiennes
- Séries de Fourier
- Calcul différentiel
- Equations différentielles
- Espaces affines
- Courbes
- Surfaces
- Intégrales multiples
- Index