

## Séries entières

- Maple dispose d'une petite librairie de manipulation de séries entières (attention à ne pas confondre avec la fonction **series** qui manipule en fait des développements limités). On le charge par
- > **with(powseries):**

### La structure des séries entières

- Pour Maple, une série entière est une fonction qui permet de calculer les coefficients, autrement dit une fonction  $n \mapsto a_n$ . On dispose de deux fonctions de créations de séries entières. La première est la procédure **powcreate** qui prend en argument une équation et des conditions initiales. L'équation définit à la fois le nom de la série entière (par exemple  $s$ ), le nom de la variable d'indice (par exemple  $n$ ), le nom du coefficient d'indice  $n$  étant alors nécessairement  $s(n)$ . L'équation doit alors être de la forme  $s(n) = \text{expr}$  où  $\text{expr}$  est une expression Maple dépendant de  $n$ , la définition pouvant être récursive. Les conditions initiales sont de la forme  $s(0) = \text{expr}_0$ , etc (elles sont indispensables si la définition du coefficient est récursive, sinon elles ont priorité sur la définition générale). Voici par exemple deux moyens de définir la série

entière  $\sum_{n=0}^{\infty} \frac{x^n}{n!}$ , l'une de nom  $s1$ , l'autre de nom  $s2$

- > **powcreate(s1(n)=1/n!);**
- > **powcreate(s2(n)=s2(n-1)/n, s2(0)=1);**
- Remarquons que Maple ne calcule pas formellement les coefficients d'une série entière

> **s1(n);**  
 $s1(n)$

> **s2(n);**  
 $s2(n)$

- Par contre, il calcule numériquement un coefficient donné

> **s1(10);**  
 $\frac{1}{3628800}$

> **s2(10);**  
 $\frac{1}{3628800}$

- Maple peut transformer une série entière en développement limité à l'aide de la fonction **tpsform** (le tps est là pour *truncate power series*), qui prend en argument la série entière, la variable souhaitée et l'ordre de troncature:

> **tpsform(s1, x, 10);**  
 $1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 + \frac{1}{720}x^6 + \frac{1}{5040}x^7 + \frac{1}{40320}x^8 + \frac{1}{362880}x^9 + O(x^{10})$

> **tpsform(s2, z, 4);**  
 $1 + z + \frac{1}{2}z^2 + \frac{1}{6}z^3 + O(z^4)$

- Inversement, la fonction **powpoly** transforme un polynôme en série entière:

> **s3:=powpoly(1+x+x^2, x);**  
 $s3 := \text{proc}(\text{powparm}) \dots \text{end}$

> **s3(1);**  
1

> **s3(3);**  
0

### Opérations sur les séries entières

- Les opérations usuelles (addition, soustraction, multiplication, quotient, puissance) peuvent être

utilisées à condition d'en forcer l'évaluation à l'aide de la fonction evalpow

```
> powcreate(s1(n)=1/n!);
powcreate(s2(n)=-1/n, 's2(0)'=0);
s3:=evalpow(s1+2*s2): tpsform(s3,z);
```

$$1 - z - \frac{1}{2}z^2 - \frac{1}{2}z^3 - \frac{11}{24}z^4 - \frac{47}{120}z^5 + O(z^6)$$

Bien entendu, Maple peut intégrer ou dériver les séries entières

```
> s3:=powdiff(s2): tpsform(s3,z);
```

$$-1 - z - z^2 - z^3 - z^4 - z^5 + O(z^6)$$

```
> s4:=powint(s2): tpsform(s4,s);
```

$$-\frac{1}{2}s^2 - \frac{1}{6}s^3 - \frac{1}{12}s^4 - \frac{1}{20}s^5 + O(s^6)$$

On peut réaliser des opérations de composition de séries entières, soit à l'aide de séries entières quelconques

```
> s3:=compose(s1,s2): tpsform(s3,z,15);
```

$$1 - z + O(z^{15})$$

soit à l'aide de séries prédéfinies; c'est ainsi que l'on peut calculer l'exponentielle, le logarithme, le sinus, le cosinus ou la racine carrée d'une série entière (dans des limites de validité)

```
> s3:=powexp(s2): tpsform(s3,z,15);
```

$$1 - z + O(z^{15})$$

```
> s3:=powlog(s1): tpsform(s3,z,15);
```

$$z + O(z^{15})$$

```
> s3:=powlog(s2);
Error, (in powseries/powlog) first term of power series has zero coefficient
```

## ■ Définition de séries classiques

Pour terminer, voici quelques définitions de séries classiques, que le lecteur pourra compléter à sa guise

```
> SEexp:=powexp(powpoly(z,z)):
SEcos:=powcos(powpoly(z,z)):
SEsin:=powsin(powpoly(z,z)):
SEtg:=evalpow(SEsin/SEcos):
SEch:=evalpow((powexp(powpoly(z,z))+powexp(powpoly(-z,z)))/2):
:
SEsh:=evalpow((powexp(powpoly(z,z))-powexp(powpoly(-z,z)))/2):
:
SEth:=evalpow(SEsh/SEch):
SElog:=powexp(powpoly(1+z,z)): # log(1+x)
```

avec une petite application classique

```
> s:=evalpow(compose(SEsin,SEtg)-compose(SEtg,SEsin)):
tpsform(s,z,15);
```

$$-\frac{1}{30}z^7 - \frac{29}{756}z^9 - \frac{1913}{75600}z^{11} - \frac{95}{7392}z^{13} + O(z^{15})$$

La série de Taylor de la fonction f peut quant à elle être définie par

```
> Taylor:=f->powcreate(`Taylor_`.f(n)=(D@@n)(f)(0)/n!):
```

Après l'appel de Taylor(nom), on dispose de la série entière Taylor\_nom :

```
> Taylor(g1);
> Taylor_g1(3);
```

$$\frac{1}{6}(D^{(3)})(g1)(0)$$

Ceci permet de reconstruire des séries entières classiques par une autre méthode (les calculs étant alors certainement plus longs)

```
> Taylor(sin);
```

```
> tpsform(Taylor_sin,z);
```

$$z - \frac{1}{6}z^3 + \frac{1}{120}z^5 + O(z^6)$$

## Sommations de séries entières

Les méthodes de sommation de Maple, appliquées à des sommes infinies, peuvent permettre de sommer des séries entières mathématiques (mais pas des séries entières Maple au sens où elles sont définies ci dessus)

```
> sum(x^n/n!,n=0..infinity);
```

$$e^x$$

```
> sum((-1)^n*x^n/n,n=1..infinity);
```

$$-\ln(1+x)$$

```
> sum((n^3-n+1)*x^n,n=0..infinity);
```

$$-\frac{-1+3x-9x^2+x^3}{(x-1)^4}$$

```
> sum((n+1)/n!*x^n,n=0..infinity);
```

$$e^x(1+x)$$

Pourtant, Maple peut échouer dans des cas élémentaires

```
> sum((n^2+1)/n!*x^n,n=0..infinity);
```

$$\sum_{n=0} \frac{(n^2+1)x^n}{n!}$$

# Cours de mathématiques

par Denis Monasse

Ed. Vuibert

## Table des matières

- Plan général
- Algèbre générale
- Algèbre linéaire
- Réduction des endomorphismes
- Topologie des espaces métriques
- Espaces vectoriels normés
- Comparaison des fonctions
- Suites et séries numériques
- Fonctions d'une variable réelle
- Intégration
- Suites et séries de fonctions

- Séries entières
- Formes quadratiques
- Formes hermitiennes
- Séries de Fourier
- Calcul différentiel
- Equations différentielles
- Espaces affines
- Courbes
- Surfaces
- Intégrales multiples
- Index